Line Search Algorithm for Optimization

- Mathe 3 (CES)
- WS20
- Lambert Theisen (`theisen@acom.rwth-aachen.de`)

```
using PlutoUI, Calculus, Gadfly, LinearAlgebra
```

Define Objective

$$f(x) = x^2$$

```
f = #1 (generic function with 1 method)
```

```
f = (x -> x[1]^2)
```

Line Search

1. Given $x^{(0)}$
2. For $k + 0, 1, 2, \ldots$ do
    1. Update: $x^{(k+1)} = x^{(k)} + \alpha_k d^{(k)}$
3. End

```
line_search (generic function with 1 method)
```

```
function line_search(f, x0, α, d, kmax)
    x = x0
    hist = []
    push!(hist, x)
    for k=1:kmax
        x = x + α(x) * d(x)
        push!(hist, x)
    end
    return x, hist
end
```

Check Line Search

- Observe that different step sizes change the result!

```
(0, Any[1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0])
```

```
line_search(f, 1, (x->1), (x->-sign(x)), 10)
```

```
(1, Any[1, -1, 1, -1, 1, -1, 1, -1, 1, -1, 1])
```

```
line_search(f, 1, (x->2), (x->-sign(x)), 10)
```

Gradient Descent

- Is basically line search with $d^{(k)} = -\nabla f(x^{(k)})$

```
hessian (generic function with 9 methods)
```

```
begin
    # some notation
    ∇ = derivative
    ∇² = hessian
end
```

```
gradient_descent (generic function with 1 method)
```

```
function gradient_descent(f, x0, α, kmax)
    return line_search(f, x0, α, (x->-∇(f, x)), kmax)
end
```

Check Gradient Descent

```
(2.03704e-10, Any[1, 0.8, 0.64, 0.512, 0.4096, 0.32768, 0.262144, 0.209715, 0.167
```

```
gradient_descent(f, 1, (x->0.1), 100)
```

```
(2.03704e-10, Any[1, -0.8, 0.64, -0.512, 0.4096, -0.32768, 0.262144, -0.209715, 0
```

```
gradient_descent(f, 1, (x->0.9), 100) # slower, oscillating but converging
```

Newton's Method for Optimization

- Is line search with $d^{(k)} = -[\nabla^2 f(x^{(k)})]^{-1} \nabla f(x^{(k)})$

```
newton (generic function with 1 method)
```

```
function newton(f, x0, α, kmax)
    return line_search(f, x0, α, (x->-inv(∇²(f, x))*∇(f, x)), kmax)
end
```

Check Newton's Method

```
(1.05879e-22, Any[1.0, -7.28306e-7, 1.05879e-22, 1.05879e-22, 1.05879e-22, 1.05879e
```

```
newton(f, 1., (x->1.0), 100) # works well 😎
```

```
(1.26764e30, Any[1.0, -2.0, 4.00001, -7.99999, 16.0, -31.9999, 63.9998, -128.0, 2
```

```
newton(f, 1., (x->3.0), 100) # diverged 😓
```

Visualize Results




```
begin
    res_n = newton(f, 1., (x->0.4), 5)
    Gadfly.plot(
        Guide.title("Newton Algorithm"),
        layer(f, minimum(res_n[2]), maximum(res_n[2])),
        layer(x=res_n[2], y=f.(res_n[2]), label=string.(1:length(res_n[2])),
Geom.point, Geom.path, Geom.label, Theme(default_color=color("red")))
    )
end
```




```
begin
    res_gd = gradient_descent(f, 1., (x->0.4), 5)
    Gadfly.plot(
        Guide.title("Gradient Descent Algorithm"),
        layer(f, minimum(res_gd[2]), maximum(res_gd[2])),
        layer(x=res_gd[2], y=f.(res_gd[2]), label=string.(1:length(res_gd[2])),
Geom.point, Geom.path, Geom.label, Theme(default_color=color("red")))
    )
end
```

Two-Dimensional Optimization

Define Objective

$$g(x, y) = x^2 + y^2$$

g = #23 (generic function with 1 method)

```
g = (x->x[1]^2+x[2]^2)
```

Check Methods

- both work

res_gd_2d =

(Float64[4.18545e-22, 4.18545e-22], Any[Float64[1.0, 1.0], Float64[0.2, 0.2], Float

```
res_gd_2d = gradient_descent(g, [1.,1.], (x->0.4), 100)
```

```
res_n_2d =
  (Float64[2.33348e-22,  3.04878e-22],  Any[Float64[1.0,  1.0],  Float64[0.6,  0.6],  Float
```

```
res_n_2d = newton(g, [1.,1.], (x->0.4), 100)
```

```
Float64[2.33348e-22,  3.04878e-22]
```

```
res_n_2d[2][end]
```

```
true
```

```
norm(res_n_2d[2][end] - [0,0]) < eps(Float64) # is converged to machine-precision?
```

Test Gradient Descent vs Newton for 2D Rosenbrock

```
begin
    ENV["MPLBACKEND"]="Agg"
    using PyPlot
end
```




```
let
    X=collect(-2:Δ:2)
    Y=collect(-1:Δ:3)
    ff = (x->x[1]^2+x[2]^2)
    F=[h([X[j],Y[i]]) for i=1:length(X), j=1:length(Y)]
    clf()
    surf(X, Y, F, cmap=:summer)
    PyPlot.title("Rosenbrock Function")
    gcf()
end
```




```
begin
    # Rosenbrock function with x* = [a,a^2], f(x*)=0
    a = 1
    b = 100
    h = (x -> (a-x[1])^2 + b*(x[2]-x[1]^2)^2)

    x0 = [-1.,0.]

    # Gradient Descent
    res_gd_2d_rb = gradient_descent(h, x0, (x->0.002), 20)
    res_gd_2d_rb_x = [res_gd_2d_rb[2][i][1] for i=1:length(res_gd_2d_rb[2])]
    res_gd_2d_rb_y = [res_gd_2d_rb[2][i][2] for i=1:length(res_gd_2d_rb[2])]

    # Newton
    res_n_2d_rb = newton(h, x0, (x->0.5), 20)
    res_n_2d_rb_x = [res_n_2d_rb[2][i][1] for i=1:length(res_n_2d_rb[2])]
    res_n_2d_rb_y = [res_n_2d_rb[2][i][2] for i=1:length(res_n_2d_rb[2])]

    clf()
    Δ = 0.1
    X=collect(-2:Δ:2)
    Y=collect(-1:Δ:3)
    F=[h([X[j],Y[i]]) for i=1:length(X), j=1:length(Y)]
    contourf(X,Y,F, levels=50)
    PyPlot.title("Rosenbrock: Gradient Descent vs. Newton")

    # res_gd_2d_rb
    PyPlot.plot(res_gd_2d_rb_x, res_gd_2d_rb_y, color="yellow")
    scatter(res_gd_2d_rb_x, res_gd_2d_rb_y, color="yellow")
    for i=1:length(res_gd_2d_rb_x)
        annotate(string(i), [res_gd_2d_rb_x[i], res_gd_2d_rb_y[i]], color="w",
zorder=2)
    end

    # res_n_2d_rb
    PyPlot.plot(res_n_2d_rb_x, res_n_2d_rb_y, color="red")
    scatter(res_n_2d_rb_x, res_n_2d_rb_y, color="red")
    for i=1:length(res_n_2d_rb_x)
```

```
        annotate(string(i), [res_n_2d_rb_x[i], res_n_2d_rb_y[i]], color="w",
zorder=2)
    end

    legend(["Gradient Descent", "Newton"])

    xlabel("x")
    ylabel("y")

    # Mark minimum
    scatter(a, a^2, color="r", s=500, zorder=3, marker="x")

    gcf()
end
```